

# Object Programmer's Guide

Randy Hollines  
object@gmail.com

February 9, 2012

## **Abstract**

An introduction to the Object programming language and its features. This article is intended to introduce programmers and compiler enthusiasts to the unique features and design of the Object programming language. Unless otherwise noted, this article covers functionality that is included as part of release *2.1.6*. For additional information please refer to the general and technical project websites.

# Contents

<b>1</b>	<b>Introduction</b>	<b>5</b>
<b>2</b>	<b>Getting Started</b>	<b>5</b>
2.1	Compiling Source . . . . .	5
2.2	Executing . . . . .	6
<b>3</b>	<b>The Basics</b>	<b>6</b>
3.1	Variable Declarations . . . . .	7
3.2	Expressions . . . . .	8
3.2.1	Mathematical and Logical Expressions . . . . .	8
3.2.2	Arrays . . . . .	9
3.2.3	Conditional Expressions . . . . .	10
3.3	Statements . . . . .	10
3.3.1	If Statement . . . . .	11
3.3.2	Select Statement . . . . .	11
3.3.3	While Statement . . . . .	11
3.3.4	Do/While Statement . . . . .	12
3.3.5	For Statements . . . . .	12
3.3.6	Each Statements . . . . .	12
<b>4</b>	<b>User Defined Types</b>	<b>13</b>
4.1	Enums . . . . .	13
4.2	Classes . . . . .	13
4.2.1	Class Inheritance . . . . .	13
4.2.2	Class Casting and Identification . . . . .	14
4.2.3	Methods and Functions . . . . .	15
4.3	Interfaces . . . . .	16
4.4	Higher-Order Functions . . . . .	16
4.4.1	Assigning and Passing Functions . . . . .	16
4.4.2	Invoking Functions . . . . .	17
<b>5</b>	<b>Native Shared Library Support</b>	<b>17</b>
<b>6</b>	<b>Debugger</b>	<b>18</b>
6.1	Starting the Debugger . . . . .	19
6.2	Debugging Commands . . . . .	19
<b>7</b>	<b>Class Libraries</b>	<b>20</b>
7.1	Core Libraries . . . . .	20
7.1.1	Base . . . . .	20
7.1.2	Bool . . . . .	20
7.1.3	Char . . . . .	20
7.1.4	Byte/Int . . . . .	21
7.1.5	Float . . . . .	22
7.1.6	String . . . . .	23

7.1.7	Bool	25
7.2	Introspection	25
7.2.1	Class	26
7.2.2	Method	26
7.2.3	DataType	26
7.2.4	TypeId	27
7.3	Data Structs	27
7.3.1	Compare	27
7.3.2	BasicCompare	27
7.3.3	List/IntList/FloatList	28
7.3.4	Stack/IntStack/FloatStack	29
7.3.5	Vector/CompareVector/IntVector/FloatVector	30
7.3.6	Map/IntMap/FloatMap/StringMap	31
7.3.7	Hash/StringHash	32
7.4	System Libraries	32
7.4.1	Runtime	32
7.4.2	Console	33
7.4.3	Thread	33
7.4.4	ThreadMutex	34
7.4.5	Date	34
7.4.6	Timer	35
7.4.7	File	35
7.4.8	FileReader	36
7.4.9	FileWriter	37
7.4.10	Directory	37
7.4.11	Serializer	38
7.4.12	Deserializer	38
7.4.13	TCP Socket	39
7.4.14	HttpClient	40
7.5	XML Libraries	41
7.5.1	XMLParser	41
7.5.2	XmlBuilder	42
7.5.3	XmlElement	42
7.5.4	RegEx	44
<b>8</b>	<b>Examples</b>	<b>45</b>
8.1	Prime Numbers	45
8.2	Simple HTTP client	45
8.3	XML Parsing and Querying	46
<b>9</b>	<b>Appendix A: Example Debugging Session</b>	<b>47</b>
9.1	Sample Source	47
9.2	Compiling the Source and Starting the Debugger	48
9.3	Setting a Breakpoint and Running the Program	48
9.4	Printing a Value	48

<b>10 Appendix B: Compiler and VM Design</b>	<b>49</b>
10.1 Compiler . . . . .	49
10.1.1 Scanner and Parser . . . . .	50
10.1.2 Contextual Analyser . . . . .	50
10.1.3 Intermediate Code Generator and Optimzier . . . . .	50
10.1.4 Target Emitter . . . . .	50
10.2 Virtual Machine . . . . .	50
10.2.1 Loader . . . . .	50
10.2.2 Interpreter . . . . .	51
10.2.3 JIT Compiler . . . . .	51
10.2.4 Memory Manager . . . . .	51
<b>11 Appendix C: VM Instruction Set</b>	<b>51</b>

# 1 Introduction

The Object program language is an object-oriented computer language with functional features. The language was designed to be an easy to use general purpose programming system. The Object language allows programmers to quickly create solutions by leveraging packaged class libraries. The syntax for the language was designed with symmetry in mind and enforces the notion that there should one intuitive way to do things. Features of the include:

- Support for object-oriented programming (all data types are treated as objects)
- Functional support (higher-order functions)
- Cross platform (support for OS X, Linux and Windows)
- Concurrent runtime JIT support for Intel processors (x86 and x64)
- Multi-threaded memory management (garbage collector)
- Local optimizations and method inlining
- Support for static libraries
- Command line debugger

## 2 Getting Started

The Object distribution consists of a compiler, virtual machine, debugger and binary inspection tool. The compiler binary is named `obc`, while the runtime virtual machine (VM) binary is named `obr`. Here is the world famous “Hello World” program written in the Object language:

```
bundle Default {  
  class Hello {  
    function : Main(args : String[]) ~ Nil {  
      "Hello World!"->PrintLine();  
    }  
  }  
}
```

### 2.1 Compiling Source

The example below compiles the source program `hello.obs` into the target binary file `hello.obe`. The two output file types that the compiler supports are executables and shared libraries. Shared libraries are binary files that contain all of the metadata needed by the compiler to relink them into executables. Both executables and shared libraries contain enough metadata to support runtime

introspection. As a naming convention, executables must end in `*.obe` while shared libraries must end in `*.obl`.

Below is an example of compiling the “Hello World” program:

```
obc -src tests\hello.obs -dest hello.obe
```

Here’s a more advanced example of linking in two required class libraries to an executable.

```
obc -src examples\xml_parser.obs -lib struct.obl,xml.obl -dest a.obe
```

Additional compiler options are:

<i>Option</i>	<i>Description</i>
<b>-src</b>	path to source files, delimited by the ‘,’ character
<b>-lib</b>	path to library files, delimited by the ‘,’ character
<b>-tar</b>	target output <b>exe</b> for executable and <b>lib</b> for library; default is <b>exe</b>
<b>-opt</b>	optimization level <b>s0–s3</b> with <b>s3</b> being the most aggressive; default is <b>s0</b>
<b>-dest</b>	output file name
<b>-debug</b>	if set, produces debug out for use by the interactive debugger (see below)

## 2.2 Executing

The command-line example below executes the `hello.obe` executable. Note, for executables all required libraries are statically linked in the target output file. When compiling shared libraries, other shared libraries are not linked into the target output library file.

```
obr hello.obe
```

## 3 The Basics

Now lets introduce the core features of the Object programming language.

Let’s first start with a list of keywords that exist in language. These words are reserved for the language and may not be used as variable, class or enum identifiers.

<i>Keywords</i>				
and	As	Bool	break	bundle
Byte	Char	class	critical	do
each	else	enum	false	Float
for	from	function	if	Int
interface	label	method	native	New
or	other	Parent	private	public
return	select	static	true	TypeOf
use	virtual	—	while	xor

In Object, all data types (excluding higher-order functions) are treated as objects. Basic objects provide supports for boolean, character, byte, integer and decimal types. These basic objects can be used to create more complex user defined objects. The listing below defines the basic objects that are supported in the language:

<i>Type</i>	<i>Description</i>
<b>Char</b>	1-byte character
<b>Char []</b>	character array
<b>Bool</b>	boolean value
<b>Bool []</b>	boolean array
<b>Byte</b>	1-byte integer
<b>Byte []</b>	byte array
<b>Int</b>	4-byte integer
<b>Int []</b>	integer array
<b>Float</b>	8-byte decimal
<b>Float []</b>	decimal array
<b>Function</b>	Two 4-byte integers

As mentioned above, basic types are objects and have associated methods for each basic class type. For example:

```
13->Min(3)->PrintLine();
3.89->Sin()->PrintLine();
-22->Abs()->PrintLine();
Float->Pi()->PrintLine();
```

### 3.1 Variable Declarations

Variables can be declared for all of the basic types described above and for user defined objects. Variables can be declared anywhere in a program and are bound to traditional block scoping rules. Variable assignments can be made during a declaration or at any other point in a program. Variables may be declared as local, instance or class level scope. Class level variables are declared using the **static** keyword. A class that is derived from another class may access it's parents variables if the parent class is declared in one of the source programs. *If a class is derived from a class declared in a shared library then that class cannot access it's parents variables, unless an accessor method is provided.* Local variables can be declared without specifying their data type, such variables are bound to the right-hand side of their first assignment. Three different declaration styles are shown below:

```
a : Int;
b : Int := 13;
c := 7;
```

Types that are not initialized at declaration time are initialized with the following default values:

<i>Type</i>	<i>Initialization</i>
<b>Char</b>	'\0'
<b>Byte</b>	0
<b>Int</b>	0
<b>Float</b>	0.0
<b>Array</b>	Nil
<b>Object</b>	Nil
<b>Function</b>	Nil

## 3.2 Expressions

The Object language supports various type of expressions. Some of these expressions include mathematical, logical, arrays and method calls. The preceding sections describes some of the expressions that are supported in Object.

### 3.2.1 Mathematical and Logical Expressions

The following code example demonstrates two ways to printing the number 42. The first way invokes the `PrintLine()` method for the literal 42. The second prints the product of a variable and a literal.

```
bundle Default {
  class Test {
    function : Main() ~ Nil {
      42->PrintLine();
      eight := 8;
      (eight * 7)->PrintLine();
    }
  }
}
```

The following mathematical operators are supported in the Object language for integers and decimal types:

- addition (+)
- subtraction (-)
- multiplication (\*)
- division (/)
- modulus – (%) - for integer values only)
- shift left – (<< - for integer values only)



- shift right – (>> - for integer values only)

In addition the following assignment operators are supported:

- addition-equals (+=)
- subtraction-equals (-=)
- multiplication-equals (\*=)
- division-equals (/=)

The following bitwise operators are also supported for integer types:

- and (**and**)
- or (**or**)
- xor (**xor**)

The `[*, /, %]` operators have a higher precedence than the `[+, -]` operators are naturally enforced by the language. Operators of the same precedence are evaluated from left-to-right. Logical operations are of lower precedence than mathematical operations. All logical operators are of the same precedence and order is determined via left-to-right evaluation. The `[&, |]` logical operators use short-circuit logic; meaning that some expressions may not be executed if evaluation criteria is not satisfied.

The following logical operators are supported in the Objeck language:

- and (**&**)
- or (**|**)
- equal (**=**)
- not-equal (**<>**)
- less-than (**<**)
- greater-than (**>**)
- less-than-equal (**<=**)
- greater-than-equal (**>=**)

### 3.2.2 Arrays

Objeck supports single and multi-dimensional arrays. Arrays are allocated dynamically from the system heap. The memory that is allocated for arrays is managed automatically by the garbage collector. All of the basic types described above (as well as user defined types) can be allocated as arrays. The code example below shows how a two-dimensional array of type `Int` is allocated and dereferenced.

```
array := Int->New[2,3];
array[0,2] := 13;
array[1,0] := 7;
```

The size of an array can be obtained by calling the array's `Size()` method. The `Size()` method will return the number of elements in a given array. For a multi-dimensional array the size method returns the number of elements in the first dimension. Character array literals are allocated as `String` objects. The `String` class provides support for advanced string operations.

```
string := "Hello World!";
string->Size()->PrintLine();
strings := ["Hello","World!"];
strings->Size()->PrintLine();
```

The following example allocates an array of `Widget` objects. An object must implement its `New` method if it's going to be instantiated.

```
class Widget {
  New() {
  }
}

class MakeWidgets {
  function : Main(args : String[]) ~ Nil {
    widgets := Widget->New[1000];
    each(i : widgets) {
      widgets[i] := Widget->New();
    };
  }
}
```

### 3.2.3 Conditional Expressions

There's also support for ternary conditional expressions. For example the following statement prints the value `false`.

```
a := 7;
b := 13;
(((a < 13) ? 10 : 20) > 15)->PrintLine();
```

## 3.3 Statements

Besides providing support for declaration statements the language has support for conditional and control statements. As with other languages, control statements can be nested in order to provide finer grain logical control. General control statements include `if` and `select` statements. Basic looping statements include `while`, `do/while`, `for` and `each` loops. Note, all statements end with the `;` character.

### 3.3.1 If Statement

An **if** statement is a control statement that executes an associated block of code if it evaluates to **true**. If the evaluation statement does not evaluate to **true** than an **else if** statement may be evaluated (if it exists), otherwise an **else** statement will be executed (if it exists). The example below demonstrates an **if** statement.

```
value := Console->ReadLine()->ToInt();
if(value <> 3) {
    "Not equal to 3"->PrintLine();
}
else if(value < 13) {
    "Less than 13"->PrintLine();
}
else {
    "Some other number"->PrintLine();
};
```

### 3.3.2 Select Statement

A **select** statement maps a value to 1 or more labels. Labels are associated to statement blocks. A label may either be a literal or an **enum** value. Multiple labels can be mapped to the same statement block. Below is an example of a **select** statement.

```
select(v) {
    label Color->Red: {
        "Red"->PrintLine();
    }

    label 9:
    label 19: {
        v->PrintLine();
    }

    label 27: {
        (3 * 9)->PrintLine();
    }
};
```

### 3.3.3 While Statement

A **while** statement is a control statement that will continue to execute it's main body as long as it's conditional expression evaluates to **true**. When its conditional expression evaluates to **false** than the loop body will cease to execute.

```

i := 10;
while(i > 0) {
    i->PrintLine();
    i -= 1;
}

```

### 3.3.4 Do/While Statement

A **do/while** statement is a control statement that will execute it's main body at least once and continue to execute it's main body as long as its conditional expression evaluates to **true**. When it's conditional expression evaluates to **false** than the loop body will cease to execute.

```

i := 10;
do {
    i->PrintLine();
    i -= 1;
}
while(i > 0);

```

### 3.3.5 For Statements

The **for** statement is another common looping construct. The **for** loop consists of a pre-condition statement followed by an evaluation expression and an update statement.

```

location := "East Bay"->ToCharArray();
for(i := 0; i < location->Size(); i += 1;) {
    location[i]->PrintLine();
}

```

### 3.3.6 Each Statements

The **each** statement is a specialized version of a **for** statement. The **each** loop consists of a counter variable and a data structure that has a **Size** method, such as arrays and **Vector** classes. The statement iterates thru all elements in the data structure.

```

values := Int->New[3];
values[0] := 5;
values[1] := 1;
values[2] := 0;

each(i : values) {
    values[i]->PrintLine();
};

```

## 4 User Defined Types

### 4.1 Enums

Enums are user defined enumerated types. The main use of an **enum** is to group a class of countable values, for example colors, into a distinct category. Once **enum** values have been defined they may not be assigned or associated to a other **enum** groups or integer classes. The valid operations for enums are as follows:

- assignment (**:=**)
- equal (**=**)
- not-equal (**<>**)

In addition, enum values may be used in **select** statements as conditional tests or labels.

```
enum Color {  
    Red,  
    Black,  
    Green  
}  
  
enum Type := -100 {  
    TYPES,  
    CLASSES,  
    INT,  
    FLOAT  
}
```

### 4.2 Classes

Classes are user defined types that allow programmers to create specialized data types. Classes are made up of attributes (data) and operations (methods). Classes are used to encapsulate programming logic and localize information. Operations that are associated to a class may either be at the class level or instance level. Class instances are created by calling an object's **New()** function. Note, an object instance can only be created if one or more **New()** functions have been defined.

#### 4.2.1 Class Inheritance

Classes may be derived from other classes using the **from** keyword. Class inheritance allows classes to share common functionality. The Objectk language supports single class inheritance, meaning that a derived class may only have one parent. The language also supports virtual classes, which assures that derived classes have been defined for all required operations declared in the base

class. Virtual classes also allow the programmer to define non-virtual methods that contain program behavior. Virtual classes are dynamically bound to implementation classes at runtime.

```
class Foo {
  @lhs : Int;

  New(lhs : Int) {
    @lhs := lhs;
  }

  method : native : AddTwo(rhs : Int) ~ Int {
    return 2 + rhs;
  }

  method : virtual : AddThree(int rhs) ~ Int;

  method : GetLhs() ~ Int {
    return lhs;
  }
}

class Bar from Foo {
  New(value : Int) {
    Parent(value);
  }

  method : native : AddThree(rhs : Int) ~ Int {
    return 3 + rhs;
  }

  function : Main() ~ Nil {
    bar : b := Bar->New(31);
    b->AddThree(9)->PrintLine();
  }
}
```

#### 4.2.2 Class Casting and Identification

An object that is inherited from another object may be either upcasted or downcasted. Object casting can be performed using the `As()` operator. The Object language detects upcasting and downcasting at compile time. Upcasting requires a runtime check, while down casting does not. If cross casting is detected then a compile time error will be generated.

```
method : public : Compare(right : Base) ~ Int {
  if(right <> Nil) {
```

```

if(GetClassID() = right->GetClassID()) {
    a := right->As(A);

    if(@value = a->Get()) {
        return 0;
    };
...

```

To determine if an object instance is of a certain class type (object or interface) it's `TypeOf` method can be invoked. This method will return `true` if the instance is of the same or a derived type, `false` otherwise. This method can be used to check a class instance type before casting it.

```

s := "FooBar";
t := s->TypeOf(String);
t->PrintLine();

```

The class that a given object instance belongs to can be found by calling its `GetClassID` method. This method returns an enum that is associated with that instance's class type. This method is generally used to determine if two object instances are of the same or different classes.

#### 4.2.3 Methods and Functions

The Object language supports both methods and functions. Functions are public static procedures that may be executed by any class. Methods are operations that may be performed on an object instance. Methods have `public` and `private` qualifiers. Methods that are `private` may only be called from within the same class, while `public` methods may be called from other classes. Note, methods are `private` by default. The Object language supports polymorphic methods and functions, meaning that there can be multiple methods with the same name within the same class as long as their declaration arguments vary.

Methods and functions can either be executed in an interpreted or JIT compiled mode. Interpreted execution mimics microprocessor functions in a platform independent manner. JIT execution takes the compiled stack code and produces native machine code. Note, that there is initial overhead involved in the JIT compilation process since it occurs at runtime. In addition, some methods can not be compiled into native machine code but this is a rare case. The keyword `native` is used to JIT compile methods and functions at runtime.

A function or method may be defined as `virtual` meaning that any class that originates from that class must implement all of the class's `virtual` methods or functions. `Virtual` methods are a way to ensure that certain operations are available to a family of classes. If a class declares a `virtual` method then the class becomes `virtual`, meaning that it cannot be directly instantiated.

Below is an example of declaring a virtual method:

```

method : virtual : public : GetMake() ~ String;

```

### 4.3 Interfaces

As a modern object-oriented language, Object supports interfaces. Interfaces define virtual methods that must be implemented by a given class. A class may implement one or more interfaces. Interface references may be passed, dereferenced and casted in a similar manner to class references. Below is an example of an interface definition:

```
interface Color {
  method : virtual : public : GetColor() ~ String;
}

interface Vehicle {
  method : virtual : SetName(name : String) ~ Nil;
  method : virtual : public : GetNumberOfWheels() ~ Int;
}

class Ufo implements Vehicle, Color {
  method : public : GetColor() ~ String {
    ...
  }
  method : SetName(name : String) ~ Nil {
    ...
  }
  method : public : GetNumberOfWheels() ~ Int {
    ...
  }
}
```

### 4.4 Higher-Order Functions

The Object language supports the notion of higher-order functions such that a given function may be bound to a variable at runtime. Variables are assigned based upon functional prototypes. Prototypes enforce strong type checking by ensuring that a function parameters and return type are consistent between assignments. Once a variable is bound, it may be assigned to other variables, passed to other functions/methods, returned from other functions/method or dynamically evoked. Please note, methods are not treated as higher-order constructs only functions.

#### 4.4.1 Assigning and Passing Functions

The following example shows how a function is defined and assigned to a variable:

```
class Foo {
  function : GetSize(s : String) ~ Int {
```



```

        return s->Size();
    }
}

....
s1 : (String) ~ Int := Foo->GetSize(String) ~ Int;
s2 := Foo->GetSize(String) ~ Int;
size1 := EnvokeSize("Hello", s1);
size2 := EnvokeSize("Hello", s2);

```

#### 4.4.2 Envoking Functions

The following example shows two ways that a function variable can be assigned and envoked:

```

...
method : public : EnvokeSize(s : String, f : (String) ~ Int) ~ Int {
    return f(s);
}
...

```

## 5 Native Shared Library Support

The Object Language has the ability to interact with native C/C++ shared libraries (shared libraries/DLLs) via runtime extensions. The APIs allow a programmer to load a shared libraries and invoke native C functions. Data is passed between the two layers via `Object[]`. Basic objects such as `Int` and `Float` types must be wrapped in `IntHolder` and `FloatHolder` classes respectively. Please refer to the examples in the source code distribution for additional information.

A native function signature looks likes like the following:

```

#ifdef _WIN32
    __declspec(dllexport)
#endif
void odbc_connect(VMContext& context);

```

The `VMContext` structure contains pointers to the calculation stack as well as utility functions that allow programmers to allocate VM objects and arrays. In addition, this structure provides the ability to invoke VM functions and methods..

```

struct VMContext {
    long* data_array;
    long* op_stack;
    long* stack_pos;
    APITools_AllocateArray_Ptr alloc_array;
    APITools_AllocateObject_Ptr alloc_obj;
    APITools_MethodCall_Ptr call_method_by_name;
}

```

```
APITools_MethodCallId_Ptr call_method_by_id;
};
```

The `lib.api.h` header file also includes helper functions that allow programmers to access and set data that passed into the native C function. The following functions are available:

- `APITools_GetFunctionValue`
- `APITools_SetFunctionValue`
- `APITools_GetIntValue`
- `APITools_GetIntAddress`
- `APITools_SetIntValue`
- `APITools_GetFloatValue`
- `APITools_GetFloatAddress`
- `APITools_GetStringValue`
- `APITools_SetStringValue`
- `APITools_CallMethod`
- `APITools_PushInt`
- `APITools_PushFloat`
- `APITools_GetArraySize`
- `APITools_SetIntArrayElement`
- `APITools_GetFloatArrayElement`
- `APITools_SetFloatArrayElement`

## 6 Debugger

The Objectc compiler toolset contains a simple interactive read-only debugger, which allows programmers to inspect values within their programs. The debugger allows programmers to set breakpoints within methods based upon source line numbers. The debugger can also calculate simple arithmetic expressions involving variables and constants. The following commands are currently supported:

## 6.1 Starting the Debugger

The source program must be compiled with the `-debug` set. The command line debugger is started up running the `odb` executable. The `-exe` option must be present and specify the path to the executable. The `-src` option is optional and specifies the path to the program source. Also note, to print instance level variables the path must start with `@self→`.

## 6.2 Debugging Commands

<i>Command</i>	<i>Description</i>	<i>Example</i>
<code>[b]reak</code>	sets a breakpoint	<code>b</code> <code>b hello.obs:10</code>
<code>breaks</code>	shows all breakpoints	
<code>[d]elete</code>	deletes a breakpoint	<code>d hello.obs:10</code>
<code>clear</code>	clears all breakpoints	
<code>[n]ext</code>	moves to the next line within the same method/function with debug information	
<code>[s]tep</code>	moves to the next line with debug information	
<code>[j]ump</code>	jumps out of an existing method/function and moves to the next line with debug information	
<code>args</code>	specifies program arguments	<code>args "Hello World"</code>
<code>[r]un</code>	runs a loaded program	
<code>[p]rint</code>	prints the value of an expression, along with meta-data	<code>p loc1.value</code> <code>p @self→inst.value</code>
<code>[l]ist</code>	lists a range of lines in a source file or the lines near the current breakpoint	<code>l</code> <code>l hello.obs:10</code>
<code>[i]nfo</code>	displays the variables for a class	<code>i</code> <code>i class=Foo</code>
<code>stack</code>	displays the method/function call stack	
<code>exe</code>	loads a new executable	<code>exe "../test.obe"</code>
<code>src</code>	specifies a new source path	<code>src "../../"</code>
<code>[q]uit</code>	exits a given debugging session	

## 7 Class Libraries

Object includes class libraries that provides access to system resources, such as files and sockets, while also providing support for basic data structures like lists and vectors. As new class libraries are added they will be documented in this section.

### 7.1 Core Libraries

Class support for basic datatypes.

#### 7.1.1 Base

Base class for all objects.

- **GetClass** - returns metadata about a class
  - method : public : `GetClass()` ~Class
- **GetClassID** - returns the class ID
  - method : public : native : `GetClassID()` ~ClassID
- **GetInstanceID** - returns a unique instance ID
  - method : public : native : `GetInstanceID()` ~Int

#### 7.1.2 Bool

- **Print** - prints the current value
  - method : native : `Print()` ~Nil
- **PrintLine** - prints the current value along with a line return
  - method : native : `PrintLine()` ~Nil
- **ToString** - converts the current value to a **String** object instance
  - method : native : `ToString()` ~String

#### 7.1.3 Char

- **ToUpper** - coverts a character to a uppercase character
  - method : public : native : `ToUpper()` ~Char
- **ToLower** - coverts a character to a lowercase character
  - method : public : native : `ToLower()` ~Char
- **IsDigit** - determines if the character is a digit (in the range of 0-9)

- method : native : IsDigit() ~Bool
- **IsChar** - determines if the character is a alpha (in the range of A-Z or a-z)
  - method : native : IsChar() ~Bool
- **Min** - returns the smallest of the two numbers; returns the same number if they are equal
  - method : native : Min(r : Char) ~Char
- **Max** - returns the largest of the two numbers; returns the same number if they are equal
  - method : native : Max(r : Char) ~Char
- **Print** - prints the current value
  - method : native : Print() ~Nil
- **PrintLine** - prints the current value along with a line return
  - method : native : PrintLine() ~Nil
- **ToString** - converts the current value to a **String** object instance
  - method : native : ToString() ~String

#### 7.1.4 Byte/Int

- **Min** - returns the smallest of the two numbers; returns the same number if they are equal
  - method : native : Min(r : Byte) ~Byte
- **Max** - returns the largest of the two numbers; returns the same number if they are equal
  - method : native : Max(r : Byte) ~Byte
- **Abs** - returns the absolute value of the current number
  - method : native : Abs() ~Byte
- **Print** - prints the current value
  - method : native : Print() ~Nil
- **PrintLine** - prints the current value along with a line return
  - method : native : PrintLine() ~Nil
- **ToString** - converts the current value to a **String** object instance
  - method : native : ToString() ~String

### 7.1.5 Float

- **Min** - returns the smallest of the two numbers; returns the same number if they are equal
  - method : native : `Min(r : Byte) ~Float`
- **Max** - returns the largest of the two numbers; returns the same number if they are equal
  - method : native : `Max(r : Byte) ~Float`
- **Abs** - returns the absolute value of the current number
  - method : native : `Abs() ~Float`
- **Floor** - returns the floor of the current number
  - method : `Floor() ~Float`
- **Ceiling** - returns the ceiling of the current number
  - method : `Ceiling() ~Float`
- **Sin** - returns the sine of the radian value
  - method : `Sin() ~Float`
- **Cos** - returns the cosine of the radian value
  - method : `Cos() ~Float`
- **Tan** - returns the tangent of the radian value
  - method : `Tan() ~Float`
- **Log** - returns the natural log of the radian value
  - method : `Log() ~Float`
- **Pi** - returns the value of Pi
  - function : `Pi() ~Float`
- **Power** - returns the exponential power value
  - method : `Power() ~Float`
- **SquareRoot** - returns the square root
  - method : `SquareRoot() ~Float`
- **Random** - returns a pseudo-random number between 0.0 and 1.0
  - function : `Random() ~Float`

- **Print** - prints the current value
  - method : native : Print() ~Nil
- **PrintLine** - prints the current value along with a line return
  - method : native : PrintLine() ~Nil
- **ToString** - converts the current value to a **String** object instance
  - method : native : ToString() ~String

### 7.1.6 String

- **New**
  - New()
  - New(s : String)
  - New(a : Char[])
  - New(a : Byte[])
- **Append** - Appends a **String**, **Char[]**, **Char**, **Int** or **Float** to the current String instance
  - method : public : native : Append(s : String) ~Nil
  - method : public : native : Append(c : Char) ~Nil
  - method : public : native : Append(b : Byte) ~Nil
  - method : public : native : Append(i : Int) ~Nil
  - method : public : native : Append(f : Float) ~Nil
  - method : public : native : Append(a : Char[]) ~Nil
- **Find** - returns the index of the first occurrence of a given Character; -1 if not found
  - method : public : native : Find(c : Char) ~Int

**Find** - returns the index of the first occurrence of a given Character; -1 if not found

  - method : public : native : Find(offset : Int, c : Char) ~Int
- **Find** - returns the index of the first occurrence of a given String; -1 if not found
  - method : public : native : Find(s : String) ~Int
  - method : public : native : Find(s : String, offset : Int) ~Int
- **Replace** - find and replace a single string instance.

- method : public : native : Replace(find : String, replace : String) ~String
- **ReplaceAll** - find and replace all string instances.
  - method : public : native : ReplaceAll(find : String, replace : String) ~String
- **Size** - returns the size of the String
  - method : public : native : Size() ~Int
- **IsEmpty** - returns true if the string is empty, false otherwise
  - method : public : native : IsEmpty() ~Bool
- **Get** - returns the Character at the given index or -1 if not found
  - method : public : native : Get(i : Int) ~Char
- **ToCharArray** - converts a string to a Char[]
  - method : public : native : ToCharArray() ~Char[]
- **ToInt** - converts a string to a Int
  - method : public : native : ToInt() ~Int
- **ToFloat** - converts a string to a Float
  - method : public : native : ToFloat() ~Int
- **SubString** - creates a new string that contains a subset of the string's contents
  - method : public : native : SubString(offset : Int) ~String
  - method : public : native : SubString(offset : Int, length : Int) ~String
- **Trim** - removes all leading and trailing whitespace
  - method : public : native : Trim() ~String
- **Pop** - removes the last character from a string
  - method : public : native : Pop() ~Char
- **StartsWith** - returns true if String starts with matching pattern; returns false otherwise
  - method : public : native : StartsWith() ~Bool
- **EndsWith** - returns true if String ends with matching pattern; returns false otherwise



- method : public : native : EndsWith() ~Bool
- **ToUpper** - coverts all lowercase characters to uppercase characters
  - method : public : native : ToUpper() ~String
- **ToLower** - coverts all uppercase characters to lowercase characters
  - method : public : native : ToLower() ~String
- **Reverse** - returns a string with reversed characters
  - method : public : native : Reverse() ~String
- **Equals** - compares two string returns **true** if they are equal
  - method : public : Equals(rhs : String) ~Bool
- **Compare** - compares two string returns 0 if they are equal
  - method : public : native : Compare(rhs : Compare) ~Int
- **Split** - breaks a string into tokens based upon a given delimiter
  - method : public : native : Split(delim : String) ~String[]
- **Print** - prints the current value
  - method : native : Print() ~Nil
- **PrintLine** - prints the current value along with a line return
  - method : native : PrintLine() ~Nil

#### 7.1.7 Bool

- **Print** - prints the current value
  - method : native : Print() ~Nil
- **PrintLine** - prints the current value along with a line return
  - method : native : PrintLine() ~Nil
- **ToString** - converts the current value to a **String** object instance
  - method : native : ToString() ~String

## 7.2 Introspection

Supports the ability to access program information at runtime.

### 7.2.1 Class

- **New**(c : String)
  - method : public : IsLoaded() ~Bool
- **IsLoaded** - returns true if this class has been loaded
  - method : public : IsLoaded() ~Bool
- **IsLoaded** - returns true if this class has been loaded, false otherwise
  - method : public : IsLoaded() ~Bool
- **GetName** - returns the name of the class
  - method : public : GetName() ~String
- **GetMethods** - returns an array of methods
  - method : public : GetMethods() ~GetMethods[]
- **GetMethodNumber** - returns the number of methods in the class
  - method : public : GetMethodNumber() ~Int

### 7.2.2 Method

- **GetClass** - returns class associated with the method
  - method : public : GetClass() ~Class
- **GetName** - returns the pretty name of the method
  - method : public : GetName() ~String
- **GetParameters** - returns the datatypes of the parameters
  - method : public : GetParameters() ~DataType[]
- **GetReturn** - returns the datatype of the return
  - method : public : GetReturn() ~DataType

### 7.2.3 DataType

- **GetType** - returns the type ID
  - method : public : GetType() ~TypeId
- **GetDimension** - returns dimensions of the datatype
  - method : public : GetDimension() ~Int
- **GetClassName** - returns string name of non-basic objects, Nil otherwise
  - method : public : GetClassName() ~String

#### 7.2.4 **TypeId**

- **BOOL**
- **BYTE**
- **CHAR**
- **INT**
- **FLOAT**
- **CLASS**
- **FUNC**

### 7.3 **Data Structs**

Provides support for basic data structures such as vectors, lists, hashes, maps, etc. These classes are located in the ‘Struct’ bundle.

#### 7.3.1 **Compare**

Interface for all objects that use comparative algorithms.

- **Compare** - compares two object instances; 0 if equal, less-than 0 if less and greater-than 0 if greater.
  - method : virtual : public : Compare(rhs : System.Compare) ~Int
- **HashID** - returns a unique value for the given class type
  - method : virtual : public : HashID() ~Int

#### 7.3.2 **BasicCompare**

Basic implementation of **Compare** methods.

- **Compare** - returns true if argument is the same instance, false otherwise.
  - method : public : Compare(rhs : System.Compare) ~Int
- **HashID** - hash value based upon instance ID
  - method : public : HashID() ~Int

### 7.3.3 List/IntList/FloatList

The List class allow values to be added, removed and deleted from a list. There are two specialized version of this class: **IntList** and **FloatList**. The **IntList** and **FloatList** classes use **Int** and **Float** types respectively instead of **Compare** type. The general class supports the following operations:

- **AddBack** - adds a new value to the back of the list
  - method : public : native : AddBack(value : Compare) ~Nil
- **RemoveBack** - removes the last element in the list
  - method : public : RemoveBack() ~Nil
- **AddFront** - adds a new value to the front of the list
  - method : public : native : AddFront(value : Compare) ~Nil
- **RemoveFront** - removes the first element in the list
  - method : public : RemoveFront() ~Nil
- **Find** - finds an element in the list
  - method : public : Find(value : Compare) ~Bool
- **Has** - returns true if item is in list
  - method : public : Has(value : Compare) ~Bool
- **Insert** - inserts a new value in the position pointed to the cursor
  - method : public : Insert(value : Compare) ~Bool
- **Remove** - removes the last element in the list
  - method : public : Remove() ~Nil
- **Insert** - inserts an element into the current cursor position
  - method : public : Insert(value : Compare) ~Bool
- **Next** - advances the internal cursor by one element
  - method : public : Next() ~Nil
- **Previous** - retreats the internal cursor by one element
  - method : public : Previous() ~Nil
- **Get** - returns the value of the element pointed to by the cursor
  - method : public : Get() ~Compare

- **Forward** - moves the cursor to the end of the list
  - method : public : Forward() ~Nil
- **Rewind** - moves the cursor to the start of the list
  - method : public : Rewind() ~Nil
- **IsStart** - returns true if cursor is at the start of the list
  - method : public : IsStart() ~Bool
- **IsEnd** - returns true if cursor is at the end of the list
  - method : public : IsEnd() ~Bool
- **Size** - returns the size of the list

#### 7.3.4 Stack/IntStack/FloatStack

The Stack class support the concept of a growing stack. There are two specialized version of this class: **IntStack** and **FloatStack**. The **IntStack** and **FloatStack** classes use **Int** and **Float** types respectively instead of **Base** type. The general class supports the following operations:

- **Push** - pushes a new value onto the stack
  - method : public : Push(value : Base) ~Nil
- **Pop** - pops a values from the top of the stack
  - method : public : Pop() ~Base
- **Top** - retrieves the top value from the stack (without popping the stack)
  - method : public : Top() ~Base
- **IsEmpty** - returns true if stack is empty
  - method : public : IsEmpty() ~Bool
- **Size** - returns the stack size
  - method : public : Size() ~Int

### 7.3.5 Vector/CompareVector/IntVector/FloatVector

The Vector class support the concept of a growing array. There are three specialized version of this class: **CompareVector**, **IntVector** and **FloatVector**. The **IntVector**, **FloatVector** and **CompareVector** classes use **Int**, **Float** and **Compare** types respectively instead of **Base** type. The general class supports the following operations:

- **New**
  - **New()**
  - **New(values : Base[])**
  - **New(values : Vector)**
- **AddBack** - adds a new value to the back of the vector
  - method : public : **AddBack(value : Base) ~Nil**
- **RemoveBack** - removes the last element in the vector
  - method : public : **RemoveBack() ~Nil**
- **Get** - returns the value of the element pointed to by the cursor
  - method : public : **Get(index : Int) ~Base**
- **Set** - replaces the list value based upon the given index
  - method : public : **Set(value : Base, index : Int) ~Bool**
- **Size** - returns the size of the list
  - method : public : **Size() ~Int**
- **ToArray** - returns an array of elements
  - method : public : **ToArray() ~Base**

Additional methods for **CompareVector/IntVector/FloatVector**

- **Sort** - sorts a vector of values using a merge sort algorithm in  $O(n \log_2 n)$  time.
  - method : public : **Sort() ~Nil**
- **BinarySearch** - performs a binary search for an element in a *sorted* vector using an iterative binary search algorithm in  $O(\log_2 n)$  time.
  - method : public : **BinarySearch(v : Compare) ~Nil**

Additional methods for **IntVector/FloatVector**

- **Min** - returns the smallest value in the vector

- method : public : native : Min() ~Int/Float
- **Max** - returns the largest value in the vector
  - method : public : native : Max() ~Int/Float
- **Average** - returns the calculated average for all values in the list
  - method : public : native : Average() ~Int/Float
- **Filter** - applies the result of the passed in function to each element in the array
  - method : public : Filter(func : (Int/Float/Compare) ~ Bool) ~IntVector/FloatVector/CompareVector
- **Apply** - applies the result of the passed in function to each element in the array
  - method : public : Apply(func : (Int/Float) ~ Int/Float), IntVector/FloatVector

### 7.3.6 Map/IntMap/FloatMap/StringMap

The Map class supports the concept of an associative array with key/value pairs. The class implements a balance binary tree algorithm such that inserts, deletes and searches are  $O(\log_2 n)$ .

- **New**
  - New()
- **Insert** - adds a new value to the tree
  - method : public : Insert(key : Compare, value : Base) ~Nil
- **Remove** - removes a value from the tree
  - method : public : Remove(key : Compare) ~Nil
- **Find** - searches for a value based upon a key
  - method : public : Find(key : Compare) ~Base
- **Has** - returns true if item is in map
  - method : public : Has(value : Compare) ~Bool
- **GetKeys** - returns a vector of keys
  - method : public : GetKeys() ~Vector
- **GetValues** - returns a vector of values

- method : public : GetValues() ~Vector
- **Size** - returns the size of the map
  - method : public : Size() ~Int

### 7.3.7 Hash/StringHash

The Hash class supports the concept of a hash table with key/value pairs. The class implements a hashing algorithm that is optimized for pairs of 256 elements or less (consider using the **Map** class for larger sets):

- **Insert** - adds a new value to the hash table
  - method : public : Insert(key : Compare, value : Base) ~Nil
- **Remove** - removes a value from the hash table
  - method : public : Remove(key : Compare) ~Nil
- **Find** - searches for a value based upon a key
  - method : public : Find(key : Compare) ~Base
- **Has** - returns true if item is in hash
  - method : public : Has(value : Compare) ~Bool
- **GetKeys** - returns a vector of keys
  - method : public : GetKeys() ~Vector
- **GetValues** - returns a vector of values
  - method : public : GetValues() ~Vector
- **Size** - returns the size of the Hash
  - method : public : Size() ~Int

## 7.4 System Libraries

These classes provide access to system resources.

### 7.4.1 Runtime

The Runtime class provides information about the runtime VM environment.

- **GetPlatform** - returns a string containing information about the runtime environment
  - function : GetPlatform() ~String



- **GetDate** - returns the current time
  - function : public : GetDate() ~Date
- **Exit** - exits the current running program
  - function : public : Exit() ~Nil
- **Copy** - copies a block of memory from one array to another
  - function : Copy(dest : Char[], dest\_offset : Int, src : Char[], src\_offset : Int, len : Int) ~Bool
  - function : Copy(dest : Int[], dest\_offset : Int, src : Int[], src\_offset : Int, len : Int) ~Bool
  - function : Copy(dest : Float[], dest\_offset : Int, src : Float[], src\_offset : Int, len : Int) ~Bool
  - function : Copy(dest : Base[], dest\_offset : Int, src : Base[], src\_offset : Int, len : Int) ~Bool
  - function : Copy(dest : Compare[], dest\_offset : Int, src : Compare[], src\_offset : Int, len : Int) ~Bool

#### 7.4.2 Console

The Console class allows programmers to read and write information to the system console. The class supports the following operations:

- **Print** - prints all basic types including **String** and **Char[]** to standard out.
  - function : Print(t : type) ~ConsoleIO
- **PrintLine** - prints all basic types including **String** and **Char[]** to standard out followed by a newline.
  - function : PrintLine(t : type) ~ConsoleIO
- **ReadString** - reads in a line of text as a **Char[]** from standard in.
  - method : public : ReadString() ~String

#### 7.4.3 Thread

The Thread class allows programmers to invoke a method in a separately running system thread. This class is in the ‘Concurrency’ bundle.

- **New**
  - New(name : String)

- **Execute** - Execute a **Thread()** by calling the thread's **Run()** method in a separate thread.
  - method : public : Execute(param : System.Base) ~Nil
- **Sleep** - Sleeps a given thread for a number of milliseconds.
  - function : Sleep(t : Int) ~Nil
- **Join** - Joins a thread instance with the creating thread
  - method : public : Join() ~Nil
- **GetName** - Returns the name of the thread
  - method : public : GetName() ~String
- **GetRunID** - Returns running thread ID
  - method : public : GetName() ~String
- **Run** - Method that is executed by the new thread
  - method : virtual : public : Run(param : System.Base) ~Nil

#### 7.4.4 ThreadMutex

The ThreadMutex class is used for thread synchronization i.e. locking and unlocking running threads.

- **New**
  - New(name : String)
- **GetName** - Returns the name of the thread
  - method : public : GetName() ~String

#### 7.4.5 Date

The Date class allows programmers gain access to the current system time. This class is in the 'Time' bundle. The class supports the following operations:

- **New**
  - New()
- **GetDay** - return the current day as an **Int**.
  - method : public : GetDay() ~Int
- **GetMonth** - return the current month as an **Int**.
  - method : public : GetMonth() ~Int

- **GetYear** - return the current year as an **Int**.
  - method : public : GetYear() ~Int
- **GetHours** - return the current hour as an **Int**.
  - method : public : GetHours() ~Int
- **GetMinutes** - return the current minutes as an **Int**.
  - method : public : GetMinutes() ~Int
- **GetSeconds** - return the current seconds as an **Int**.
  - method : public : GetSeconds() ~Int
- **IsSavingsTime** - return true if daylight saving time, false otherwise
  - method : public : IsSavingsTime() ~Bool

#### 7.4.6 Timer

The Time class allows programmers perform simple timings. This class is in the ‘Time’ bundle. The class supports the following operations:

- **New**
  - New()
- **Start** - Starts the timing
  - method : public : Start() ~Nil
- **End** - Ends the timing
  - method : public : End() ~Nil
- **End** - Returns the elapsed time in milliseconds
  - method : public : GetElapsedTime() ~Int

#### 7.4.7 File

The File class allows programmers manipulate system files. This class is in the ‘IO’ bundle. The class supports the following operations:

- **New**
  - New(name : String)
- **IsOpen** - returns true if file is open.
  - method : public : IsOpen() ~Bool

- **IsEOF** - returns true if the file pointer is at the EOF.
  - method : public : IsEOF() ~Bool
- **Seek** - seeks to a position in a file.
  - method : public : Seek(p : Int) ~Bool
- **Rewind** - moves the file pointer to the beginning of a file.
  - method : public : Rewind() ~Nil
- **Size** - returns the size of the file.
  - function : Size(name : String) ~Int
- **Remove** - deletes a file.
  - function : Remove(n : String) ~Bool
- **Exists** - returns true if the file exists.
  - function : Exists(n : String) ~Bool
- **Rename** - renames a file
  - function : Rename(from : String, to : String) ~Bool

#### 7.4.8 FileReader

The `FileReader` is inherited from the `File` class and allows programmers read files. This class is in the ‘IO’ bundle. The class supports the following operations:

- **New**
  - New(name : String)
- **Close** - closes a file.
  - method : public : Close() ~Nil
- **ReadByte** - reads a byte from a file.
  - method : public : ReadByte() ~Byte
- **ReadBinaryFile** - reads an entire file into a byte array.
  - function : public : ReadBinaryFile(file : String) ~Byte[]
- **ReadBuffer** - reads n number of bytes from a file.
  - method : public : ReadBuffer(offset : Int, num : Int, buffer : Byte[]) ~Int

- **ReadString** - reads a line from a file.
  - method : public : ReadString() ~String
- **ReadFile** - reads an entire file.
  - method : public : ReadFile() ~String

#### 7.4.9 FileWriter

The **FileReader** is inherited from the **File** class and allows programmers read files. This class is in the 'IO' bundle. The class supports the following operations:

- **Close** - closes a file.
  - method : public : Close() ~Nil
- **WriteByte** - writes a byte to a file.
  - method : public : WriteByte(b : Int) ~Bool
- **WriteBuffer** - writes n number of bytes to a file.
  - method : public : WriteBuffer(buffer : Byte[]) ~Int
  - method : public : WriteBuffer(offset : Int, num : Int, buffer : Byte[]) ~Int
- **WriteString** - writes a string to a file.
  - method : public : WriteString(s : String) ~Nil

#### 7.4.10 Directory

The **Directory** class allows programmers manipulate file system directories. This class is in the 'IO' bundle. The class supports the following operations:

- **Create** - creates a new directory.
  - function : Create(n : String) ~Bool
- **Exists** - returns true if the directory exists.
  - function : Exists(n : String) ~Bool
- **List** - returns vector of file and directory names.
  - function : List(n : String) ~String[]

#### 7.4.11 Serializer

The Serializer class allows datatypes to be deflated into a byte array for storage or transmission. This class is in the ‘IO’ bundle. The class supports the following operations:

- **New**
  - New()
- **Add** - deflates a datatype
  - method : public : Write(b : Bool) ~Nil
  - method : public : Write(i : Int) ~Nil
  - method : public : Write(f : Float) ~Nil
  - method : public : Write(o : Base) ~Nil
  - method : public : Write(b : Bool[]) ~Nil
  - method : public : Write(b : Bool[]) ~Nil
  - method : public : Write(b : Byte[]) ~Nil
  - method : public : Write(c : Char[]) ~Nil
  - method : public : Write(i : Int[]) ~Nil
  - method : public : Write(f : Float[]) ~Nil
- **Serialize** - returns a byte stream of serialized data
  - method : public : Serialize() ~Byte[]

#### 7.4.12 Deserializer

The Deserializer class allows datatypes to be inflated from a byte array. This class is in the ‘IO’ bundle. The class supports the following operations:

- **New**
  - New(b : Byte[])
- **ReadBool** - reads a boolean value
  - method : public : ReadBool() ~Bool
- **ReadInt** - reads an integer value
  - method : public : ReadInt() ~Int

- **ReadFloat** - reads a float value
  - method : public : ReadFloat() ~Float
- **ReadObject** - reads an object
  - method : public : ReadObject() ~Float
- **ReadBoolArray** - reads a boolean array
  - method : public : ReadBoolArray() ~Bool[]
- **ReadByteArray** - reads a byte array
  - method : public : ReadByteArray() ~Byte[]
- **ReadCharArray** - reads a character array
  - method : public : ReadCharArray() ~Char[]
- **ReadIntArray** - reads an integer array
  - method : public : ReadIntArray() ~Byte[]
- **ReadFloatArray** - reads a float array
  - method : public : ReadFloatArray() ~Float[]

#### 7.4.13 TCPSocket

The **TCPSocket** class allows programmers connect to TCP/IP socket servers. This class in the ‘Net’ bundle. The class supports the following operations:

- **New**
  - New(address : String, port : Int)
- **IsOpen** - returns true if the socket is connected.
  - method : public : IsOpen() ~Bool
- **Close** - closes a connected socket.
  - method : public : Close() ~Nil
- **WriteByte** - writes a byte to a file.
  - method : public : WriteByte(b : Int) ~Bool
- **WriteBuffer** - writes n number of bytes to a file.
  - method : public : WriteBuffer(offset : Int, num : Int, buffer : Byte[]) ~Int
- **WriteString** - writes a string to a file.

- method : public : WriteString(s : String) ~Nil
- **ReadByte** - reads a byte from a file.
  - method : public : ReadByte() ~Byte
- **ReadBuffer** - reads n number of bytes from a file.
  - method : public : ReadBuffer(offset : Int, num : Int, buffer : Byte[]) ~Int
- **ReadString** - reads a line from a file.
  - method : public : ReadString() ~String
- **HostName** - returns the host name
  - function : HostName() ~String

#### 7.4.14 HttpClient

The HttpClient class allows programmers access HTTP 1.1 websites. This class in the ‘Net’ bundle:

- **New**
  - New();
- **Post** - Performs a HTTP POST to the connected website, returns true is successful
  - method : public : Post(url : String, port : Int, content : String) ~Vector
  - method : public : Post(url : String, content\_type: String, port : Int, content : String) ~Vector
- **Get** - Performs a HTTP GET from the connected website, returns a Vector of strings
  - method : public : Get(url : String, port : Int) ~Vector
  - method : public : Get(url : String, content\_type: String, port : Int) ~Vector
- **GetHeaders** - Returns a Map of headers information. HTTP header information is populated after a GET and POST requests.
  - method : GetHeaders() ~Map
- **GetCookies** - Gets a list of cookies. The cookie list can be cleared using this method.
  - method : GetCookies() ~List



- **CookiesEnabled** - Enables or disables cookie support.
  - method : CookiesEnabled(state : Bool) ~List
- **SetCookie** - Sets a cookie.
  - method : SetCookie(cookie : String) ~Nil

## 7.5 XML Libraries

These classes provide support for XML parsing and document creation.

### 7.5.1 XMLParser

The XMLParser class allows programmers to parse and search XML documents using a simplified DOM structure. This class is in the ‘XML’ bundle:

- **New**
  - New(buffer : Char[])
  - New(buffer : String)
- **ToString** - serializes the DOM into a string
  - method : public : ToString() ~String
- **Parse** - Parses an XML stream into a DOM structure. Returns true if successful; false otherwise.
  - method : public : Parse() ~Bool
- **GetVersion** - returns the XML document version
  - method : public : GetVersion() ~String
- **GetEncoding** - returns the XML character encoding
  - method : public : GetEncoding() ~String
- **GetRoot** - Returns the root node of the DOM.
  - method : public : GetRoot() ~XmlElement
- **FindElements** - returns all elements that match the given query string. The following XPath expressions are supported num, attrib\_name first() and last().
  - method : public : FindElements(path : String) ~Vector
- **GetError** - Returns the last parsing error; if one occurred.
  - method : public : Parse() ~Bool

### 7.5.2 XmlBuilder

The XmlBuilder class allows a programmer to dynamically build an XML DOM. Once the DOM is build it may be traversed and/or serialized into a string. This class in the ‘XML’ bundle:

- **New**
  - New(root : String)
  - New(root : String, version : String, encoding : String)
- **GetRoot** - Returns the root element.
  - method : public : GetRoot() ~XmlElement
- **ToString** - Serializes an element into a string
  - method : public : ToString() ~String
- **GetVersion** - Returns the version name.
  - method : public : GetVersion() ~String
- **SetVersion** - Sets the version name.
  - method : public : SetVersion(name : String) ~Nil
- **GetEncoding** - Returns the encoding name.
  - method : public : GetEncoding() ~String
- **SetEncoding** - Sets the encoding name.
  - method : public : SetEncoding(name : String) ~Nil

### 7.5.3 XmlElement

The XmlElement class represents a canonical XML node within a DOM. This class in the ‘XML’ bundle:

- **New**
  - New(type : XmlElementType, name : String)
  - New(type : XmlElementType, name : String, content : String)
  - New(type : XmlElementType, name : String, attribs : StringHash, content : String)
- **GetName** - Returns the element name.
  - method : public : GetName() ~String
- **SetName** - Sets the element name.

- method : public : `SetName(name : String) ~Nil`
- **GetRoot** - Returns the root element.
  - method : public : `GetRoot() ~XmlElement`
- **GetType** - Returns the element type.
  - method : public : `GetType() ~XmlElementType`
- **SetType** - Sets the element type.
  - method : public : `SetType(type : XmlElementType) ~Nil`
- **GetParent** - Gets the parent element
  - method : public : `GetParent() ~XmlElement`
- **SetParent** - Sets the parent element
  - method : public : `SetParent(parent : XmlElement) ~Nil`
- **GetContent** - Returns the XML content between two tags.
  - method : public : `GetContent() ~String`
- **SetContent** - Sets the XML content between two tags.
  - method : public : `SetContent(content : String) ~Nil`
- **GetChildCount** - Returns number of logical child elements.
  - method : public : `GetChildCount() ~Int`
- **FindElements** - returns all elements that match the given query string.  
The following XPath expressions are supported num, attrib\_name, first() and last(). Note: current node name must be part of the query.
  - method : public : `FindElements(path : String) ~Vector`
- **GetChildren** - Returns all child elements.
  - method : public : `GetChildren() ~Vector`
- **GetChildren** - Returns all child elements that match the given name.
  - method : public : `GetChildren(name : String) ~Vector`
- **GetChild** - Returns a child element.
  - method : public : `GetChild(i : Int) ~XmlElement`
- **AddChild** - adds a child element.
  - method : public : `AddChild(child : XmlElement) ~Nil`

- **GetAttribute** - Returns the value of a given attribute.
  - method : public : GetAttribute(name : String) ~String
- **SetAttribute** - Sets the name/value of a given attribute.
  - method : public : SetAttribute(name : String, value : String), Nil
- **ToString** - Serializes an element into a string
  - method : public : ToString() ~String
- **DecodeString** - Decodes an XML encoded string
  - function : DecodeString(in : String) ~String
- **EncodeString** - Encodes an XML encoded string
  - function : EncodeString(in : String) ~String

#### 7.5.4 RegEx

The RegEx class provides support for regular expressions. This class is in *beta* status. This class in the ‘RegEx’ bundle:

- *Expressions*
  - . match any 1
  - ? match 0 or 1
  - \* match 0 or more
  - + match 1 or more
  - () grouping
  - | or in grouping
- **New** - Compilers the regular expression
  - New(pattern : String)
- **Match** - Attempts to match input string using supplied pattern. Returns true if successful, false otherwise.
  - method : public : MatchAll(input : String) ~Bool
- **Find** - Finds the last index of the matching pattern. Return -1 if expression does not match.
  - method : public : Find(input : String) ~Int
  - method : public : Find(input : String, offset : Int) ~Int

## 8 Examples

### 8.1 Prime Numbers

```
bundle Default {
  class FindPrime {
    function : Main() ~ Nil {
      Run(1000000);
    }

    function : native : Run(topCandidate : Int)~ Nil {
      candidate : Int := 2;
      while(candidate <= topCandidate) {
        trialDivisor : Int := 2;
        prime : Int := 1;

        found : Bool := true;
        while(trialDivisor * trialDivisor <= candidate & found) {
          if(candidate % trialDivisor = 0) {
            prime := 0;
            found := false;
          }
          else {
            trialDivisor := trialDivisor + 1;
          };
        };

        if(found) {
          candidate->PrintLine();
        };
        candidate := candidate + 1;
      };
    }
  }
}
```

### 8.2 Simple HTTP client

```
use Net;
use IO;
use Struct;

bundle Default {
  class HttpTest {
    function : Main(args : String[]) ~ Nil {
      if(args->Size() = 1) {
```

```

        client := HttpClient->New();
        lines := client->Get(args[0], 80);
        for(i := 0; i < lines->Size(); i := i + 1;) {
            lines->Get(i)->As(String)->PrintLine();
        };
    };
}
}
}

```

### 8.3 XML Parsing and Querying

```

use IO;
use XML;

bundle Default {
    class Test {
        function : Main(args : String[]) ~ Nil {
            if(args->Size() = 1) {
                parser := XmlParser->New(FileReader->ReadFile(args[0]));
                if(parser->Parse()) {
                    results :=
                        parser->FindElements("//nutrition/food[last()]/calories[fat]");
                    if(results <> Nil) {
                        each(i : results) {
                            element := results->Get(i)->As(XmlElement);
                            element->GetAttribute("total")->PrintLine();
                        };
                    };
                }
            }
            else {
                parser->GetError()->PrintLine();
            };
        };
    }
}
}

```

## 9 Appendix A: Example Debugging Session

### 9.1 Sample Source

```
bundle Default {  
  class Bar {  
    v1 : Float;  
    v2 : Int;  
  
    New() {  
      v1 := 2.31;  
      v2 := 26;  
    }  
  }  
  
  class Foo {  
    bar : Bar;  
    value : Int;  
  
    New(v : Int) {  
      value := v;  
    }  
  
    method : public : Get() ~ Int {  
      return value;  
    }  
  
    method : public : SetBar() ~ Nil {  
      bar := Bar->New();  
    }  
  }  
  
  class Test {  
    function : Main(args : System.String[]) ~ Nil {  
      d : Float := 11.12;  
      z := Int->New[5,6];  
      z[2,3] := 27;  
  
      f := Foo->New(24);  
      f->SetBar();  
      v := f->Get();  
    }  
  }  
}
```

The sample file is named `debug.obs`.

## 9.2 Compiling the Source and Starting the Debugger

```
obc -src test_src\debug.obs -dest a.obe -debug
obd -exe ..\..\compiler\a.obe -src ..\..\compiler\test_src
```

```
-----
Objeck v1.0.0 - Interactive Debugger
-----
```

```
loaded executable: file='../../compiler/a.obe'
source files: path='../../compiler/test_src/'
```

## 9.3 Setting a Breakpoint and Running the Program

```
> b debug.obs:31
added break point: file='debug.obs:31'
> r
break: file='debug.obs:31', method='Test->Main(..)'
> l
List
    26:  }
    27:  }
    28:
    29:  class Test {
    30:  function : Main(args : System.String[]) ~ Nil {
=>    31:  d : Float := 11.12;
    32:  z := Int->New[5,6];
    33:  z[2,3] := 27;
    34:
    35:  f := Foo->New(24);
    36:  f->SetBar();
> n
break: file='debug2.obs:32', method='Test->Main(..)'
```

## 9.4 Printing a Value

```
> p d
print: type=Float, value=11.12
> b debug.obs:37
added break point: file='debug.obs:37'
> c
break: file='debug2.obs:37', method='Test->Main(..)'
> p z
print: type=Int[], value=2197556(0x218834), dimension=2, size=30
> p z[2,3]
print: type=Int[], value=27(0x1b)
> p f->value
```




```

print: type=Int, value=24
> p f->bar
print: type=Bar, value=0x218864
> p f->bar->v1
print: type=Float, value=2.31
> q
> p f->bar->v1 * 3.5
print: type=Float, value=8.085
goodbye.

```

## 10 Appendix B: Compiler and VM Design



The following section gives a brief overview of the major architectural components that comprise the Objective-C language compiler and virtual machine.

### 10.1 Compiler

The language compiler is written in C++ and makes heavy use of the C++ STL for portability across platforms. As mentioned in the introduction, the compiler accepts source files and shared libraries as inputs and produces either executables or shared libraries. Note, the compiler has two modes of operation: **User Mode** compiles traditional end-user programs, while **System Mode** compiles system libraries and processes special system language directives.

### 10.1.1 Scanner and Parser

The scanner component reads source files and parses the text into tokens. The scanner works in conjugation with the  $LL(k)$  parser by providing  $k$  lookahead tokens for parsing. Note, the scanner can only scan system language directives while in **System Mode**. The source parser is a recursive-decent parser that generates an abstract parser tree, which is passed to the Contextual Analyser for validation.

### 10.1.2 Contextual Analyser

The Contextual Analyser is responsible for ensuring that a source program is valid. In addition, the context analyser also creates relationships between contextually resolved entities (i.e. methods  $\longleftrightarrow$  method calls). The analyser accepts an abstract parser tree and shared libraries as input and produces a decorated parse tree as output. The decorated parse tree is then passed to the Intermediate Code Generator for the production of VM instructions.

### 10.1.3 Intermediate Code Generator and Optimzier

The Intermediate Code Generator accpets a decorated parse and produces a flat list of VM stack instructions. These instruction lists are then passed to the Optimizer for basic block optimizations (constant folding, strength reduction, instruction simplification and method inlining).

### 10.1.4 Target Emitter

Finally, the improved intermediate code is passed to code emitter component, which writes it to a file.

## 10.2 Virtual Machine

The language VM is written in C/C++ and was designed to be highly portable. The VM makes heavy use of operating system specific APIs (i.e. WIN32 and POSIX) but does so in an abstracted manner. The JIT compiler is targeted to produce machine code for the IA-32 and AMD64 (future) hardware architectures.

### 10.2.1 Loader

The loader component allows the VM to read target code structures such as classes, methods and VM instructions. The loader create an in-memory representation of this information, which is used by the VM interpreter and JIT compiler. In addition, the loader processes command-line parameters that are passed into the VM prior to execution.

### **10.2.2 Interpreter**

The Interpreter executes stack based VM instructions (listed below) and manages two primary stacks: the execution stack and call stack. The execution stack is used to manage the data that is needed for VM calculations. The call stack is used to manage function/method calls and the states between those calls.

### **10.2.3 JIT Compiler**

The JIT compiler translates stack based VM instruction into processor specific machine code (i.e. IA-32). The JIT compiler is evoked by the interpreter and methods are translated in a separate execution thread. This process allow methods to be executed concurrently in an interpreted manner while they are being compiled into machine code. Note, methods are only converted into machine code once.

### **10.2.4 Memory Manager**

The Memory Manager component allows the runtime system to manage the user allocation/deallocation of heap memory. The memory managers implementes a multi-thread “mark and sweep” algorithm. The marking stage of the process is multi-thread, such that, each root is scanned in a separate thread. The sweeping stage is done in a single thread since the data structures that are needed to manage the state of the running program are modified.

## **11 Appendix C: VM Instruction Set**

The appendix below lists the types of stack instructions that are executed by the Object VM. The VM was designed to be portable and language independent. Early development versions of the VM included an inline assembler, which may be re-added in future releases.

<b>Stack Operators</b>		
<i>Mnemonic</i>	<i>Opcode(s)</i>	<i>Description</i>
LOAD_INT_LIT	4-byte integer	pushes integer onto stack
LOAD_FLOAT_LIT	8-byte float	pushes float onto stack
LOAD_INT_VAR	variable index	pushes integer onto stack
LOAD_FLOAT_VAR	variable index	pushes float onto stack
LOAD_FUNC_VAR	variable index	pushes float onto stack
LOAD_SELF	n/a	pushes self integer on stack
STOR_INT_VAR	variable index	pops integer from stack and saves to index location
STOR_FLOAT_VAR	variable index	pops float from stack and saves to index location
STOR_FUNC_VAR	variable index	pops float from stack and saves to index location
COPY_INT_VAR	variable index	copies an integer from stack and saves to index location
COPY_FLOAT_VAR	variable index	copies a float from stack and saves to index location
LOAD_BYTE_ARY_ELM	array dimension	pushes byte onto stack; assumes array address was pushed prior
LOAD_INT_ARY_ELM	array dimension	pushes integer onto stack; assumes array address was pushed prior
LOAD_FLOAT_ARY_ELM	array dimension	pushes float onto stack; assumes array address was pushed prior
LOAD_ARY_SIZE	n/a	pushes array size as integer onto stack; assumes array address was pushed prior
STOR_BYTE_ARY_ELM	variable index	stores byte at index location; assumes array address was pushed prior
STOR_INT_ARY_ELM	variable index	stores integer at index location ; assumes array address was pushed prior
STOR_FLOAT_ARY_ELM	variable index	stores float at index location; assumes array address was pushed prior

<b>Logical Operators</b>		
<i>Mnemonic</i>	<i>Opcode(s)</i>	<i>Description</i>
EQL_INT	n/a	pops top two integer values and pushes result of equal operation
NEQL_INT	n/a	pops top two integer values and pushes result of not-equal operation
LES_INT	n/a	pops top two integer values and pushes result of less-than operation
GTR_INT	n/a	pops top two integer values and pushes result of greater-than operation
LES_EQL_INT	n/a	pops top two integer values and pushes result of less-than-equal operation
GTR_EQL_INT	n/a	pops top two integer values and pushes result of greater-than-equal operation
EQL_FLOAT	n/a	pops top two floats values and pushes result of equal operation
NEQL_FLOAT	n/a	pops top two floats values and pushes result of not-equal operation
LES_FLOAT	n/a	pops top two floats values and pushes result of less-than operation
GTR_FLOAT	n/a	pops top two floats values and pushes result of greater-than operation
LES_EQL_FLOAT	n/a	pops top two floats values and pushes result of less-than-equal operation
GTR_EQL_FLOAT	n/a	pops top two floats values and pushes result of greater-than-equal operation

<b>Logical Operators</b>		
<i>Mnemonic</i>	<i>Opcode(s)</i>	<i>Description</i>
AND_INT	n/a	pops top two integer values and pushes result of and operation
OR_INT	n/a	pops top two integer values and pushes result of or operation

<b>Mathematical Operators</b>		
<i>Mnemonic</i>	<i>Opcode(s)</i>	<i>Description</i>
ADD_INT	n/a	pops top two integer values and pushes result of add operation
SUB_INT	n/a	pops top two integer values and pushes result of subtract operation
MUL_INT	n/a	pops top two integer values and pushes result of multiply operation
DIV_INT	n/a	pops top two integer values and pushes result of divide operation
SHL_INT	n/a	pops top two floats values and pushes result of shift left operation
SHR_INT	n/a	pops top two floats values and pushes result of shift right operation
MOD_INT	n/a	pops top two integer values and pushes result of modulus operation
ADD_FLOAT	n/a	pops top two floats values and pushes result of greater-than-equal operation
SUB_FLOAT	n/a	pops top two floats values and pushes result of subtract operation
MUL_FLOAT	n/a	pops top two floats values and pushes result of multiply operation
DIV_FLOAT	n/a	pops top two floats values and pushes result of divide operation
I2F	n/a	pop top integer and pushes result of float cast
F2I	n/a	pop top float and pushes result of integer cast

<b>Methods</b>		
<i>Mnemonic</i>	<i>Opcod</i> e(s)	<i>Description</i>
SWAP_INT	n/a	swaps the top two integer values on the stack
POP_INT	n/a	control pop of an integer from the stack
POP_FLOAT	n/a	control pop of a float from the stack
RTRN	n/a	exits existing method returning control to callee
MTHD_CALL	integer values for class id and method id	synchronous call to given method releasing control
DYN_MTHD_CALL	pops integer values for class id and method id	dynamic synchronous call to given method releasing control
ASYNM_MTHD_CALL	integer values for class id and method id; pushes new thread id	asynchronous call to given method
ASYNM_JOIN	thread id	waits for identified thread to end execution
LBL	label id	identifies a jump label

<b>Objects and Memory Operations</b>		
<i>Mnemonic</i>	<i>Opcode(s)</i>	<i>Description</i>
JMP	label id and conditional context (1=true, 0=unconditional, -1=false)	jump to label id
NEW_BYTE_ARRAY	array dimension	pushes address of new byte array
NEW_INT_ARRAY	array dimension	pushes address of new integer array
NEW_FLOAT_ARRAY	array dimension	pushes address of new float array
NEW_OBJ_INST	integer value for class id	pushes address of new class instance
OBJ_TYPE_OF	integer value of “check” class	performs runtime object instance check
OBJ_INST_CAST	integer values for “from” class and “to” class	performs runtime class cast check (note: only required for up casting)
CPY_BYTE_ARRAY	destination of source array, offset of destination array, address of source array, offset of source array, number of	copies elements from one array to another
CPY_INT_ARRAY	destination of source array, offset of destination array, address of source array, offset of source array, number of	copies elements from one array to another
CPY_FLOAT_ARRAY	destination of source array, offset of destination array, address of source array, offset of source array, number of	copies elements from one array to another



<b>Traps and Threads</b>		
<i>Mnemonic</i>	<i>Opcode(s)</i>	<i>Description</i>
THREAD_CREATE	n/a	creates an new thread instance (calculation stack and stack pointer)
THREAD_WAIT	n/a	waits for worker threads to stop execution
CRITICAL_START	n/a	creates a mutex such that only one thread can execute in a given section
CRITICAL_END	n/a	releases a system mutex
TRAP	integer value for trap id	calls runtime subroutine releasing control
TRAP_RTRN	integer value for trap id and number of arguments	calls runtime subroutine releasing control and then processes an integer return value
LIB_NEW_OBJ_INST	n/a	symbolic library link for a new object instance
LIB_MTHD_CALL	n/a	symbolic library link for a method call
LIB_OBJ_INST_CAST	n/a	symbolic library link for an object cast